Pensieve header: The full $sl_2$ invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

## Profiling

In[ ]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.
This version: June 2018. Original version: July 1994.
```

Out[ ]= Show Profile Monitor

In[ ]:=
```
(*Dynamic[PrintProfile[], UpdateInterval→3, TrackedSymbols→{}]*)
```

## External Utilities

In[ ]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

# Program

Program

## Internal Utilities

Program

Canonical Form:

Program

In[ ]:=
```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@ExpandDenominator@ExpandNumerator@Together[
     Expand[ℰ] //. e^x_ e^y_ :> e^(x+y) /. e^x_ :> e^CF[x]];
```

Program

The Kronecker $\delta$:

Program

In[ ]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

```
In[∘]:= 𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
          CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
        𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
        𝔼[L_, Q_, P_]$k_ := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

## Zip and Bind

Program

Variables and their duals:

Program

```
In[∘]:= {t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
        {τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)_i;
```

Program

Finite Zips:

Program

```
In[∘]:= collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
        collect[ℰ_, ζ_] := PP_Collect@Collect[ℰ, ζ];
        Zip_{}[P_] := P; Zip_{ζ_, ζs___}[P_] := PP_Zip[
          (collect[P // Zip_{ζs}, ζ] /. f_. ζ^d_. :> ∂_{ζ*, d} f) /. ζ* → 0]
```

Program

QZip implements the “$Q$-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the $L$ variables as scalars.

Program

```
In[∘]:= QZip_{ζs_List, simp_}@𝔼[L_, Q_, P_] := PP_QZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
            zs = Table[ζ*, {ζ, ζs}];
            c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
            ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
            ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
            qt = Inverse@Table[Kδ_{z, ζ*} - ∂_{z, ζ} Q, {ζ, ζs}, {z, zs}];
            zrule = Thread[zs → qt.(zs + ys)];
            Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
            simp /@ 𝔼[L, Q2, Det[qt] e^(-Q2) Zip_{ζs}[e^Q1 (P /. zrule)]]];
        QZip_{ζs_List} := QZip_{ζs, CF};
```

Program

Upper to lower and lower to Upper:

Program

```
In[∘]:= U2l = {B_i_^p_. → e^(-p ħ γ b_i), B^p_. → e^(-p ħ γ b), T_i_^p_. → e^(p ħ t_i), T^p_. → e^(p ħ t), 𝒜_i_^p_. → e^(p γ α_i), 𝒜^p_. → e^(p γ α)};
        l2U = {e^(c_. b_i_ + d_.) :> B_i^(-c/(ħ γ)) e^d, e^(c_. b + d_.) :> B^(-c/(ħ γ)) e^d,
            e^(c_. t_i_ + d_.) :> T_i^(c/ħ) e^d, e^(c_. t + d_.) :> T^(c/ħ) e^d,
            e^(c_. α_i_ + d_.) :> 𝒜_i^(c/γ) e^d, e^(c_. α + d_.) :> 𝒜^(c/γ) e^d,
            e^ℰ_ :> e^(Expand@ℰ)};
```

Program

LZip implements the “$L$-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of $Pe^Q$ as a single”$P$”. Here the $z$'s are $b$ and $\alpha$ and the $\zeta$'s are $\beta$ and $a$.

Program

```
In[]:= LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] :=
  PP_LZip@Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_ζs[e^{L1+Q1} (P /. U2l /. zrule)]] //. l2U];
LZip_{ζs_List} := LZip_{ζs,CF};
```

Program

```
In[]:= Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := PP_Bind@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
     ] // LZip_{Flatten@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{l_List}[L_, R_] := Bind_l[L, R]; B_{is___}[L_, R_] := Bind_{is}[L, R];
```

Program

```
In[]:= dom[𝔼[L_, Q_, P_]] := Union@Cases[L, (α | β | τ)_i_ :> i, ∞];
codom[𝔼[L_, Q_, P_]] := Union@Cases[L, (a | b | t)_i_ :> i, ∞];
E1_𝔼 // E2_𝔼 := E1 ~B_{codom[E1]⋂dom[E2]}~ E2;
```

Program

## "Define" code

Program

Define[lhs = rhs, ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

```
In[]:= SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_{is__} = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
   ReleaseHold[Hold[
     SD[op_{nisp,$k_Integer}, PP_{Boot@$k}@Block[{i, j, k}, op_{isp,$k} = ℰ; op_{nis,$k}]];
     SD[op_isp, op_{{is},$k}]; SD[op_{sis__}, op_{{sis}}];
    ] /. {SD → SetDelayed,
     isp → {is} /. {i → i_, j → j_, k → k_},
     nis → {is} /. {i → ii, j → jj, k → kk},
     nisp → {is} /. {i → ii_, j → jj_, k → kk_}
    }] ]]
```

Program

## Booting Up

Program

In[∘]:=
```
$k = 2; ℏ = γ = 1;
```

Program

In[∘]:=
```
Define[am_{i,j→k} = 𝔼[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k},
  bm_{i,j→k} = 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i} - 1) η_j y_k}]_{$k}]
```

Program

In[∘]:=
```
Define[R_{i,j} = 𝔼[ℏ a_j b_i, ℏ x_j y_i, e^( Sum_{k=2}^{$k+1} (1 - e^{γ ϵ ℏ})^k (ℏ y_i x_j)^k / (k (1 - e^{k γ ϵ ℏ})) )]_{$k},
  P_{i,j} = 𝔼[β_i α_j / ℏ, η_i ξ_j / ℏ,
    1 + If[$k == 0, 0, Normal@P_{{i,j},$k-1}⟦3⟧ - (R_{1,2} // ((P_{{1,j},0})_{$k} (P_{{i,2},$k-1})_{$k}))⟦3⟧]]]
```

Program

In[∘]:=
```
Define[aS_i = 𝔼[-α_i a_j, -ξ_i x_i,
      Sum[Expand[ e^{ξ_i x_i} (-ℏ γ ϵ)^k / (2^k k!) Nest[Expand[x_i^2 ∂_{{x_i,2}} #] &, e^{-ξ_i e^{ℏ ϵ a_i} x_i}, k]], {k, 0, $k}]]_{$k} //
    am_{i,j→i},
  aS̄_i = 𝔼[-a_i α_i, -x_i 𝒜_i ξ_i, 1 + If[$k == 0, 0,
      Normal@aS̄_{{i},$k-1}⟦3⟧ - ((aS̄_{{i},0})_{$k} // aS_i // (aS̄_{{i},$k-1})_{$k})⟦3⟧]]]
```

Program

In[∘]:=
```
Define[bS_i = (R_{i,1} // aS_1) ~B_1~ P_{i,1},
  bS̄_i = (R_{i,1} // aS̄_1) ~B_1~ P_{i,1},
  aΔ_{i→j,k} = (R_{1,j} R_{2,k}) // bm_{1,2→3} // P_{3,i},
  bΔ_{i→j,k} = (R_{j,1} R_{k,2}) // am_{1,2→3} // P_{i,3}]
```

Program

```
Define[dm_{i,j→k} =
    (𝔼[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1] (aΔ_{i→1,2} // aΔ_{2→2,3} // aS̄_3) (bΔ_{j→-1,-2} // bΔ_{-2→-2,-3})) //
     (P_{-1,3} P_{-3,1} am_{2,j→k} bm_{i,-2→k}),
  dS_i = 𝔼[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] // (bS̄_1 aS_2) // dm_{2,1→i},
  dΔ_{i→j,k} = (bΔ_{i→3,1} aΔ_{i→2,4}) // (dm_{3,4→k} dm_{1,2→j})]
```

Program

```
Define[R̄_{i,j} = R_{i,j} // dS_j,
  CC_i = 𝔼[0, 0, B_i^{1/2} e^{-ℏ ϵ a_i/2}]_{$k},
  CC̄_i = 𝔼[0, 0, B_i^{-1/2} e^{ℏ ϵ a_i/2}]_{$k},
  Kink_i = (R_{1,3} CC̄_2) ~B_{1,2}~ dm_{1,2→1} ~B_{1,3}~ dm_{1,3→i},
  Kink̄_i = (R̄_{1,3} CC_2) ~B_{1,2}~ dm_{1,2→1} ~B_{1,3}~ dm_{1,3→i}]
```

Program

Note. $t == \epsilon a - \gamma b$ and $b == -t/\gamma + \epsilon a/\gamma$.

Program

In[ ]:=
```
Define[b2t_i = 𝔼[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, e^(ϵ β_i a_i/γ)]_$k,
  t2b_i = 𝔼[α_i a_i - τ_i γ b_i, ξ_i x_i + η_i y_i, e^(ϵ τ_i a_i)]_$k]
```

# Testing

In[ ]:=
```
BeginProfile[]
```

Out[ ]=
```
ProfileRoot
```

In[∘]:=

$$\textbf{Block}\Big[\{\$k = 1\},\ \Big\{ \mathbf{am} \to \mathbf{am}_{i,j\to k},\ \mathbf{bm} \to \mathbf{bm}_{i,j\to k},\ \mathbf{dm} \to \mathbf{dm}_{i,j\to k},\ \mathbf{R} \to \mathbf{R}_{i,j},\ \overline{\mathbf{R}} \to \overline{\mathbf{R}}_{i,j},\ \mathbf{P} \to \mathbf{P}_{i,j},\ \mathbf{aS} \to \mathbf{aS}_i,$$
$$\overline{\mathbf{aS}} \to \overline{\mathbf{aS}}_i,\ \mathbf{bS} \to \mathbf{bS}_i,\ \overline{\mathbf{bS}} \to \overline{\mathbf{bS}}_i,\ \mathbf{dS} \to \mathbf{dS}_i,\ \mathbf{a\Delta} \to \mathbf{a\Delta}_{i\to j,k},\ \mathbf{b\Delta} \to \mathbf{b\Delta}_{i\to j,k},\ \mathbf{d\Delta} \to \mathbf{d\Delta}_{i\to j,k},$$
$$\mathbf{CC} \to \mathbf{CC}_i,\ \overline{\mathbf{CC}} \to \overline{\mathbf{CC}}_i,\ \mathbf{Kink} \to \mathbf{Kink}_i,\ \overline{\mathbf{Kink}} \to \overline{\mathbf{Kink}}_i,\ \mathbf{b2t} \to \mathbf{b2t}_i,\ \mathbf{t2b} \to \mathbf{t2b}_i \Big\}\Big] \ // \ \mathbf{Column}$$

Out[∘]=

$$am \to \mathbb{E}\left[a_k\,(\alpha_i + \alpha_j),\ x_k\left(e^{-\gamma\,\alpha_j}\,\xi_i + \xi_j\right),\ 1\right]$$

$$bm \to \mathbb{E}\left[b_k\,(\beta_i + \beta_j),\ y_k\,(\eta_i + \eta_j),\ 1 - y_k\,\beta_i\,\eta_j\,\epsilon + O[\epsilon]^2\right]$$

$$dm \to \mathbb{E}\Big[a_k\,\alpha_i + a_k\,\alpha_j + b_k\,\beta_i + b_k\,\beta_j,\ \frac{1}{\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\left(\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_i + \hbar\,y_k\,\mathcal{A}_j\,\eta_j + \hbar\,x_k\,\mathcal{A}_i\,\xi_i + \mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i - B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i + \hbar\,x_k\,\mathcal{A}_i\,\mathcal{A}_j\,\xi_j\right),$$
$$1 + \frac{1}{4\,\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\Big(-4\,\hbar\,y_k\,\mathcal{A}_j\,\beta_i\,\eta_j - 4\,\hbar\,x_k\,\mathcal{A}_i\,\beta_j\,\xi_i + 4\,\gamma\,\hbar^2\,x_k\,y_k\,\eta_j\,\xi_i + 4\,\hbar\,a_k\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i + 2\,\gamma\,\hbar\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i - 6\,\gamma\,\hbar\,B_k\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i + 2\,\gamma\,\hbar\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2 -$$
$$6\,\gamma\,\hbar\,B_k\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2 + \gamma\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 - 4\,\gamma\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 + 3\,\gamma\,B_k^2\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2\Big)\,\epsilon + O[\epsilon]^2\Big]$$

$$R \to \mathbb{E}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1 - \frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon + O[\epsilon]^2\right]$$

$$\overline{R} \to \mathbb{E}\left[-\hbar\,a_j\,b_i,\ -\frac{\hbar\,x_j\,y_i}{B_i},\ 1 + \frac{\left(-4\,\hbar^2\,a_j\,B_i\,x_j\,y_i - 3\,\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon}{4\,B_i^2} + O[\epsilon]^2\right]$$

$$P \to \mathbb{E}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1 + \frac{\gamma\,\eta_i^2\,\xi_j^2\,\epsilon}{4\,\hbar} + O[\epsilon]^2\right]$$

$$aS \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1 + \frac{1}{2}\left(-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i - \gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon + O[\epsilon]^2\right]$$

$$\overline{aS} \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1 + \frac{1}{2}\left(2\,\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i - 2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i - \gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon + O[\epsilon]^2\right]$$

$$bS \to \mathbb{E}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1 + \frac{\left(-2\,B_i\,y_i\,\beta_i\,\eta_i - \gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2} + O[\epsilon]^2\right]$$

$$\overline{bS} \to \mathbb{E}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1 + \frac{\left(2\,\gamma\,\hbar\,B_i\,y_i\,\eta_i - 2\,B_i\,y_i\,\beta_i\,\eta_i - \gamma\,\hbar\,y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2} + O[\epsilon]^2\right]$$

$$dS \to \mathbb{E}\Big[-a_i\,\alpha_i - b_i\,\beta_i,\ \frac{-\hbar\,y_i\,\mathcal{A}_i\,\eta_i - \hbar\,B_i\,x_i\,\mathcal{A}_i\,\xi_i + \mathcal{A}_i\,\eta_i\,\xi_i - B_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{\hbar\,B_i},$$
$$1 + \frac{1}{4\,\hbar\,B_i^2}\Big(4\,\gamma\,\hbar^2\,B_i\,y_i\,\mathcal{A}_i\,\eta_i - 4\,\hbar\,B_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i - 2\,\gamma\,\hbar^2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2 - 4\,\hbar^2\,a_i\,B_i^2\,x_i\,\mathcal{A}_i\,\xi_i - 4\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i -$$
$$4\,\gamma\,\hbar\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i + 4\,\hbar\,a_i\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i + 4\,\gamma\,\hbar\,B_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i - 4\,\gamma\,\hbar^2\,B_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i +$$
$$4\,B_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i - 4\,B_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i + 6\,\gamma\,\hbar\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i - 2\,\gamma\,\hbar\,B_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i - 2\,\gamma\,\hbar^2\,B_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2 +$$
$$6\,\gamma\,\hbar\,B_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2 - 2\,\gamma\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2 - 3\,\gamma\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2 + 4\,\gamma\,B_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2 - \gamma\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\Big)\,\epsilon + O[\epsilon]^2\Big]$$

$$a\Delta \to \mathbb{E}\left[a_j\,\alpha_i + a_k\,\alpha_i,\ x_j\,\xi_i + x_k\,\xi_i,\ 1 + \frac{1}{2}\left(-2\,\hbar\,a_j\,x_k\,\xi_i + \gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon + O[\epsilon]^2\right]$$

$$b\Delta \to \mathbb{E}\left[b_j\,\beta_i + b_k\,\beta_i,\ B_k\,y_j\,\eta_i + y_k\,\eta_i,\ 1 + \frac{1}{2}\,\gamma\,\hbar\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon + O[\epsilon]^2\right]$$

$$d\Delta \to \mathbb{E}\Big[a_j\,\alpha_i + a_k\,\alpha_i + b_j\,\beta_i + b_k\,\beta_i,\ y_j\,\eta_i + B_j\,y_k\,\eta_i + x_j\,\xi_i + x_k\,\xi_i,$$
$$1 + \frac{1}{2}\left(\gamma\,\hbar\,B_j\,y_j\,y_k\,\eta_i^2 - 2\,\hbar\,a_j\,x_k\,\xi_i + \gamma\,\hbar\,x_j\,x_k\,\xi_i^2\right)\epsilon + O[\epsilon]^2\Big]$$

$$CC \to \mathbb{E}\left[0,\ 0,\ \sqrt{B_i} - \frac{1}{2}\left(\hbar\,a_i\,\sqrt{B_i}\right)\epsilon + O[\epsilon]^2\right]$$

$$\overline{CC} \to \mathbb{E}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}} + \frac{\hbar\,a_i\,\epsilon}{2\,\sqrt{B_i}} + O[\epsilon]^2\right]$$

$$Kink \to \mathbb{E}\left[\hbar\,a_i\,b_i,\ \hbar\,x_i\,y_i,\ \frac{1}{\sqrt{B_i}} + \frac{\left(2\,\hbar\,a_i - \gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,\sqrt{B_i}} + O[\epsilon]^2\right]$$

$$\overline{Kink} \to \mathbb{E}\left[-\hbar\,a_i\,b_i,\ -\frac{\hbar\,x_i\,y_i}{B_i},\ \sqrt{B_i} + \frac{\left(-2\,\hbar\,a_i\,B_i^2 - 4\,\hbar^2\,a_i\,B_i\,x_i\,y_i - 3\,\gamma\,\hbar^3\,x_i^2\,y_i^2\right)\epsilon}{4\,B_i^{3/2}} + O[\epsilon]^2\right]$$

$$b2t \to \mathbb{E}\left[a_i\,\alpha_i - \frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i + x_i\,\xi_i,\ 1 + \frac{a_i\,\beta_i\,\epsilon}{\gamma} + O[\epsilon]^2\right]$$

$$t2b \to \mathbb{E}\left[a_j\,\alpha_i - \gamma\,b_j\,\tau_i,\ y_j\,\eta_i + x_j\,\xi_i,\ 1 + a_j\,\tau_i\,\epsilon + O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

In[∘]:=
```
Timing@{{"[a,x]" → ((𝔼[0, 0, a₂ x₁] ~B₁,₂~ am₁,₂→₁)⟦3⟧ - (𝔼[0, 0, a₁ x₂] ~B₁,₂~ am₁,₂→₁)⟦3⟧),
     "[b,y]" → ((𝔼[0, 0, y₂ b₁] ~B₁,₂~ bm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ b₂] ~B₁,₂~ bm₁,₂→₁)⟦3⟧)} /.
   z_1 → z,
  {"Δ[y]" → Last[𝔼[0, 0, y₁] ~B₁~ bΔ₁→₁,₂],
   "Δ[b]" → Last[𝔼[0, 0, b₁] ~B₁~ bΔ₁→₁,₂],
   "Δ[a]" → Last[𝔼[0, 0, a₁] ~B₁~ aΔ₁→₁,₂],
   "Δ[x]" → Last[𝔼[0, 0, x₁] ~B₁~ aΔ₁→₁,₂]},
  {
    "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ aS₁)⟦3⟧),
    "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ aS₁)⟦3⟧),
    "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ bS₁)⟦3⟧),
    "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ bS₁)⟦3⟧)
  } /. z_1 → z}
```

Out[∘]=
$$\{0.59375, \{\{[a,x] \to -x\gamma,\ [b,y] \to -y\epsilon + O[\epsilon]^3\},\ \{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta[x] \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$$

**Hopf algebra axioms on both sides separately.**

Associativity of am and bm:

In[∘]:=
```
Timing@Block[{$k = 3},
  HL /@ {(am₁,₂→₁ ~B₁~ am₁,₃→₁) ≡ (am₂,₃→₂ ~B₂~ am₁,₂→₁), (bm₁,₂→₁ ~B₁~ bm₁,₃→₁) ≡ (bm₂,₃→₂ ~B₂~ bm₁,₂→₁)}
 ]
```

Out[∘]= {0.109375, {True, True}}

*R* and *P* are inverses:

In[∘]:=
```
Timing@Block[{$k = 3}, {R_i,j, P_i,k, HL[R_i,j ~B_i~ P_i,k ≡ 𝔼[a_j α_k, x_j ξ_k, 1]]}]
```

Out[∘]=
$$\Big\{0.109375, \Big\{\mathbb{E}\Big[\hbar a_j b_i, \hbar x_j y_i, 1 - \frac{1}{4}(\gamma \hbar^3 x_j^2 y_i^2)\epsilon + \Big(\frac{1}{9}\gamma^2\hbar^5 x_j^3 y_i^3 + \frac{1}{32}\gamma^2\hbar^6 x_j^4 y_i^4\Big)\epsilon^2 + \frac{1}{1152}(24\gamma^3\hbar^5 x_j^2 y_i^2 - 72\gamma^3\hbar^7 x_j^4 y_i^4 - 32\gamma^3\hbar^8 x_j^5 y_i^5 - 3\gamma^3\hbar^9 x_j^6 y_i^6)\epsilon^3 + O[\epsilon]^4\Big],$$
$$\mathbb{E}\Big[\frac{\alpha_k\beta_i}{\hbar}, \frac{\eta_i\xi_k}{\hbar}, 1 + \frac{\gamma\eta_i^2\xi_k^2\epsilon}{4\hbar} + \frac{(36\gamma^2\hbar^2\eta_i^2\xi_k^2 + 40\gamma^2\hbar\eta_i^3\xi_k^3 + 9\gamma^2\eta_i^4\xi_k^4)\epsilon^2}{288\hbar^2} - \frac{1}{1152\hbar^3}(-48\gamma^3\hbar^4\eta_i^2\xi_k^2 - 192\gamma^3\hbar^3\eta_i^3\xi_k^3 - 156\gamma^3\hbar^2\eta_i^4\xi_k^4 - 40\gamma^3\hbar\eta_i^5\xi_k^5 - 3\gamma^3\eta_i^6\xi_k^6)\epsilon^3 + O[\epsilon]^4\Big], \text{True}\Big\}\Big\}$$

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

In[∘]:=
```
Timing[HL /@ {aS̄₁ ~B₁~ aS₁ ≡ 𝔼[a₁ α₁, x₁ ξ₁, 1], bS̄₁ ~B₁~ bS₁ ≡ 𝔼[b₁ β₁, y₁ η₁, 1]}]
```

Out[∘]= {0.328125, {True, True}}

(co)-associativity on both sides

In[∘]:= $\text{Timing}[\text{HL} /@ \{(a\Delta_{1\to1,2} \sim B_2 \sim a\Delta_{2\to2,3}) \equiv (a\Delta_{1\to1,3} \sim B_1 \sim a\Delta_{1\to1,2}),\ (b\Delta_{1\to1,2} \sim B_2 \sim b\Delta_{2\to2,3}) \equiv (b\Delta_{1\to1,3} \sim B_1 \sim b\Delta_{1\to1,2}),\ (am_{1,2\to1} \sim B_1 \sim am_{1,3\to1}) \equiv (am_{2,3\to2} \sim B_2 \sim am_{1,2\to1}),\ (bm_{1,2\to1} \sim B_1 \sim bm_{1,3\to1}) \equiv (bm_{2,3\to2} \sim B_2 \sim bm_{1,2\to1})\}]$

Out[∘]= {0.375, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:= $\text{Timing}[\text{HL} /@ \{am_{1,2\to1} \sim B_1 \sim a\Delta_{1\to1,2} \equiv (a\Delta_{1\to1,3}\ a\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (am_{3,4\to2}\ am_{1,2\to1}),\ bm_{1,2\to1} \sim B_1 \sim b\Delta_{1\to1,2} \equiv (b\Delta_{1\to1,3}\ b\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (bm_{3,4\to2}\ bm_{1,2\to1})\}]$

Out[∘]= {0.59375, {True, True}}

S is convolution inverse of id

In[∘]:= $\text{Timing}[\text{HL}[\# \equiv \mathbb{E}[0, 0, 1]]\ \&\ /@ \{(a\Delta_{1\to1,2} \sim B_1 \sim aS_1) \sim B_{1,2} \sim am_{1,2\to1},\ (a\Delta_{1\to1,2} \sim B_2 \sim aS_2) \sim B_{1,2} \sim am_{1,2\to1},\ (b\Delta_{1\to1,2} \sim B_1 \sim bS_1) \sim B_{1,2} \sim bm_{1,2\to1},\ (b\Delta_{1\to1,2} \sim B_2 \sim bS_2) \sim B_{1,2} \sim bm_{1,2\to1}\}]$

Out[∘]= {0.625, {True, True, True, True}}

S is an algebra anti-(co)morphism

In[∘]:= $\text{Timing}[\text{HL} /@ \{am_{1,2\to1} \sim B_1 \sim aS_1 \equiv (aS_1\ aS_2) \sim B_{1,2} \sim am_{2,1\to1},\ bm_{1,2\to1} \sim B_1 \sim bS_1 \equiv (bS_1\ bS_2) \sim B_{1,2} \sim bm_{2,1\to1},\ aS_1 \sim B_1 \sim a\Delta_{1\to1,2} \equiv a\Delta_{1\to2,1} \sim B_{1,2} \sim (aS_1\ aS_2),\ bS_1 \sim B_1 \sim b\Delta_{1\to1,2} \equiv b\Delta_{1\to2,1} \sim B_{1,2} \sim (bS_1\ bS_2)\}]$

Out[∘]= {0.859375, {True, True, True, True}}

Pairing axioms

In[∘]:= $\text{Timing}[\text{HL} /@ \{(bm_{1,2\to1}\ \mathbb{E}[\alpha_3\, a_3, \xi_3\, x_3, 1]) \sim B_{1,3} \sim P_{1,3} \equiv (\mathbb{E}[\beta_1\, b_1, \eta_1\, y_1, 1]\ \mathbb{E}[\beta_2\, b_2, \eta_2\, y_2, 1]\ a\Delta_{3\to4,5}) \sim B_{1,4} \sim P_{1,4} \sim B_{2,5} \sim P_{2,5},\ (b\Delta_{1\to1,2}\ \mathbb{E}[\alpha_3\, a_3, \xi_3\, x_3, 1]\ \mathbb{E}[\alpha_4\, a_4, \xi_4\, x_4, 1]) \sim B_{1,3} \sim P_{1,3} \sim B_{2,4} \sim P_{2,4} \equiv (\mathbb{E}[\beta_1\, b_1, \eta_1\, y_1, 1]\ am_{3,4\to3}) \sim B_{1,3} \sim P_{1,3}\}]$

Out[∘]= {0.375, {True, True}}

In[∘]:= $\text{Timing}[\text{HL} /@ \{(bS_1\ \mathbb{E}[\alpha_2\, a_2, \xi_2\, x_2, 1]) \sim B_{1,2} \sim P_{1,2} \equiv (\mathbb{E}[\beta_1\, b_1, \eta_1\, y_1, 1]\ aS_2) \sim B_{1,2} \sim P_{1,2},\ (\overline{bS}_1\ \mathbb{E}[\alpha_2\, a_2, \xi_2\, x_2, 1]) \sim B_{1,2} \sim P_{1,2} \equiv (\mathbb{E}[\beta_1\, b_1, \eta_1\, y_1, 1]\ \overline{aS}_2) \sim B_{1,2} \sim P_{1,2}\}]$

Out[∘]= {0.234375, {True, True}}

**Tests for the double.**

Check the double formulas on the generators agree with SL2Portfolio.pdf:

```
In[]:= Timing@{{
        "[a,y]" → ((𝔼[0, 0, y2 a1] ~B1,2~ dm1,2→1)⟦3⟧ − (𝔼[0, 0, y1 a2] ~B1,2~ dm1,2→1)⟦3⟧),
        "[b,x]"  → ((𝔼[0, 0, x2 b1] ~B1,2~ dm1,2→1)⟦3⟧ − (𝔼[0, 0, x1 b2] ~B1,2~ dm1,2→1)⟦3⟧),
        "xy-qyx" → ((𝔼[0, 0, x1 y2] ~B1,2~ dm1,2→1)⟦3⟧ − (1 + ϵ) (𝔼[0, 0, y1 x2] ~B1,2~ dm1,2→1)⟦3⟧)
       } /. {z_1 → z} // Expand // Factor,
     {
      "Δ(a)" → ((𝔼[0, 0, a1] ~B1~ dΔ1→1,2)⟦3⟧),
      "Δ(x)" → ((𝔼[0, 0, x1] ~B1~ dΔ1→1,2)⟦3⟧),
      "Δ(b)" → ((𝔼[0, 0, b1] ~B1~ dΔ1→1,2)⟦3⟧),
      "Δ(y)" → ((𝔼[0, 0, y1] ~B1~ dΔ1→1,2)⟦3⟧)
     } // Simplify,
     {
       "S(a)" → ((𝔼[0, 0, a1] ~B1~ dS1)⟦3⟧),
       "S(x)" → ((𝔼[0, 0, x1] ~B1~ dS1)⟦3⟧),
       "S(b)" → ((𝔼[0, 0, b1] ~B1~ dS1)⟦3⟧),
       "S(y)" → ((𝔼[0, 0, y1] ~B1~ dS1)⟦3⟧)
     } /. {z_1 → z} // Simplify
    }
```

Out[]= $\{7.75, \{\{[a,y] \to -y\gamma + O[\epsilon]^3, [b,x] \to x\epsilon + O[\epsilon]^3,$

$xy\text{-}qyx \to \left(-x y + \frac{1 - B + x y \hbar}{\hbar}\right) + (a B - x y + x y \gamma \hbar)\,\epsilon + \frac{1}{2}\left(-a^2 B \hbar + x y \gamma^2 \hbar^2\right)\epsilon^2 + O[\epsilon]^3\},$

$\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3, \Delta(x) \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3,$

$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3, \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^3\},$

$\{S(a) \to -a + O[\epsilon]^3, S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}\left(a^2 x \hbar^2\right)\epsilon^2 + O[\epsilon]^3,$

$S(b) \to -b + O[\epsilon]^3, S(y) \to -\frac{y}{B} + \frac{y\gamma\hbar\epsilon}{B} - \frac{\left(y\gamma^2\hbar^2\right)\epsilon^2}{2B} + O[\epsilon]^3\}\}\}$

(co)-associativity

```
In[]:= Timing[HL /@
        {(dΔ1→1,2 ~B2~ dΔ2→2,3) ≡ (dΔ1→1,3 ~B1~ dΔ1→1,2), (dm1,2→1 ~B1~ dm1,3→1) ≡ (dm2,3→2 ~B2~ dm1,2->1)}]
```

Out[]= {7.3125, {True, True}}

Δ is an algebra morphism

```
In[]:= Timing@HL[dm1,2→1 ~B1~ dΔ1→1,2 ≡ (dΔ1→1,3 dΔ2→2,4) ~B1,2,3,4~ (dm3,4→2 dm1,2→1)]
```

Out[]= {8.76563, True}

S is convolution inverse of id

```
In[]:= Timing[
        HL[# ≡ 𝔼[0, 0, 1]] & /@ {(dΔ1→1,2 ~B1~ dS1) ~B1,2~ dm1,2→1, (dΔ1→1,2 ~B2~ dS2) ~B1,2~ dm1,2→1}]
```

Out[]= {9.29688, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:= `Timing[HL /@ Expand /@ {dm_{1,2→1} ~ B_1 ~ dS_1 ≡ (dS_1 dS_2) ~ B_{1,2} ~ dm_{2,1→1}, dS_1 ~ B_1 ~ dΔ_{1→1,2} ≡ dΔ_{1→2,1} ~ B_{1,2} ~ (dS_1 dS_2)}]`

Out[∘]= {14.875, {True, True}}

Quasi-triangular axiom 1:

In[∘]:= `Timing@HL[R_{1,2} ~ B_1 ~ dΔ_{1->1,3} ≡ (R_{1,4} R_{3,2}) ~ B_{2,4} ~ dm_{2,4→2}]`

Out[∘]= {0.53125, True}

Quasi-triangular axiom 2:

In[∘]:= `Timing@HL[((dΔ_{1->1,2} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2})) ≡ ((dΔ_{1→2,1} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{3,1→1} dm_{4,2→2}))]`

Out[∘]= {6.14063, True}

The Drinfel'd element inverse property, $(u_1 \overline{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to 1} \equiv \mathbb{E}[0, 0, 1]$:

In[∘]:= `Timing@HL[((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i} ≡ 𝔼[0, 0, 1]]`

Out[∘]= {2.5625, True}

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv$^{-1}$. It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:= `Timing@Block[{$k = 3}, (((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) ~ B_i ~ dS_i) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i}]`

Out[∘]= $\left\{40.5156,\ \mathbb{E}\left[0,\ 0,\ \frac{1}{B_i} + \frac{\hbar\, a_i\, \epsilon}{B_i} + \frac{\hbar^2\, a_i^2\, \epsilon^2}{2\, B_i} + \frac{\hbar^3\, a_i^3\, \epsilon^3}{6\, B_i} + O[\epsilon]^4\right]\right\}$

In[∘]:= `Timing@Block[{$k = 2}, HL /@ {(CC_i \overline{CC}_j) ~ B_{i,j} ~ dm_{i,j→i} ≡ 𝔼[0, 0, 1], (\overline{CC}_i \overline{CC}_j) ~ B_{i,j} ~ dm_{i,j→i} ≡ (((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) ~ B_i ~ dS_i) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i}}]`

Out[∘]= {4.01563, {True, True}}

Reidemeister 2:

In[∘]:= `Timing[HL[# ≡ 𝔼[0, 0, 1]] & /@ {(\overline{R}_{1,2} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2}), (R_{1,2} \overline{R}_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2})}]`

Out[∘]= {5.5, {True, True}}

Cyclic Reidemeister 2:

In[∘]:= `Timing@HL[(R_{1,4} \overline{R}_{5,2} \overline{CC}_3) ~ B_{2,4} ~ dm_{2,4→2} ~ B_{1,3} ~ dm_{1,3→1} ~ B_{1,5} ~ dm_{1,5→1} ≡ \overline{CC}_1]`

Out[∘]= {1.96875, True}

Reidemeister 3:

In[∘]:= `Timing@HL[((R_{1,2} R_{4,3} R_{5,6}) ~ B_{1,4} ~ dm_{1,4→1} ~ B_{2,5} ~ dm_{2,5→2} ~ B_{3,6} ~ dm_{3,6→3}) ≡ ((R_{1,6} R_{2,3} R_{4,5}) ~ B_{1,4} ~ dm_{1,4→1} ~ B_{2,5} ~ dm_{2,5→2} ~ B_{3,6} ~ dm_{3,6→3})]`

Out[∘]= {4.85938, True}

Relations between the four kinks:

In[∘]:=
```
Timing[HL /@ {Kink_i ≡ (R_{3,1} CC_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i},
   Kinkbar_j ≡ (Rbar_{3,1} CCbar_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j}, (Kink_i Kinkbar_j) ~ B_{i,j} ~ dm_{i,j→1} ≡ 𝔼[0, 0, 1]}]
```

Out[∘]= {4.25, {True, True, True}}

---

The Trefoil

In[∘]:=
```
Timing@Block[{$k = 1},
  Z = R_{1,5} R_{6,2} R_{3,7} CCbar_4 Kinkbar_8 Kinkbar_9 Kinkbar_10;
  Do[Z = Z ~ B_{1,r} ~ dm_{1,r→1}, {r, 2, 10}];
  {Simplify /@ Z, Simplify /@ (Z ~ B_1 ~ b2t_1 /. T_1 → T)}]
```

Out[∘]=

$$\left\{10.75, \left\{\mathbb{E}\left[0, 0, \frac{B_1}{1 - B_1 + B_1^2} - \left(\hbar B_1 \left(-a_1 \left(-1 + B_1 - B_1^3 + B_1^4\right) + \gamma \left(B_1 - 2 B_1^2 - 2 B_1^4 + 2 \hbar x_1 y_1 + B_1^3 \left(3 + 2 \hbar x_1 y_1\right)\right)\right) \epsilon\right) \Big/ \left(1 - B_1 + B_1^2\right)^3 + O[\epsilon]^2\right],\right.\right.$$

$$\left.\left.\mathbb{E}\left[0, 0, \frac{T}{1 - T + T^2} + \left(T \hbar \left(T \left(-1 + 2 T - 3 T^2 + 2 T^3\right) \gamma + 2 \left(-1 + T - T^3 + T^4\right) a_1 - 2 \left(1 + T^3\right) \gamma \hbar x_1 y_1\right) \epsilon\right) \Big/ \left(1 - T + T^2\right)^3 + O[\epsilon]^2\right]\right\}\right\}$$

*Program*

In[∘]:=
```
Define[kR_{i,j} = R_{i,j} ~ B_{i,j} ~ (b2t_i b2t_j) /. t_{i|j} → t,
  kRbar_{i,j} = Rbar_{i,j} ~ B_{i,j} ~ (b2t_i b2t_j) /. t_{i|j} → t,
  km_{i,j→k} = (t2b_i t2b_j) ~ B_{i,j} ~ dm_{i,j→k} ~ B_k ~ b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
  kCC_i = CC_i ~ B_i ~ b2t_i /. T_i → T,
  kCCbar_i = CCbar_i ~ B_i ~ b2t_i /. T_i → T,
  kKink_i = Kink_i ~ B_i ~ b2t_i /. {t_i → t, T_i → T},
  kKinkbar_i = Kinkbar_i ~ B_i ~ b2t_i /. {t_i → t, T_i → T}]
```

*Trefoil*

In[∘]:=
```
Timing@Block[{$k = 1},
  Z = kR_{1,5} kR_{6,2} kR_{3,7} kCCbar_4 kKinkbar_8 kKinkbar_9 kKinkbar_10;
  Do[Z = Z ~ B_{1,r} ~ km_{1,r→1}, {r, 2, 10}];
  Simplify /@ Z]
```

*Trefoil*

Out[∘]=

$$\left\{4.375, \mathbb{E}\left[0, 0, \frac{T}{1 - T + T^2} + \left(T \hbar \left(T \left(-1 + 2 T - 3 T^2 + 2 T^3\right) \gamma + 2 \left(-1 + T - T^3 + T^4\right) a_1 - 2 \left(1 + T^3\right) \gamma \hbar x_1 y_1\right) \epsilon\right) \Big/ \left(1 - T + T^2\right)^3 + O[\epsilon]^2\right]\right\}$$

RVK, rot, Z from 2016-09/OneSmidgen.nb.

In[∘]:=
```
RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing sites
    are indexed 1 through 2n, and rots⟦k⟧ is the rotation between
    site k-1 and site k. RVK is also a casting operator converting
    to the RVK presentation from other knot presentations.";
RVK[pd_PD] := Module[{n, xs, x, rots, front, k},
   n = Length[pd];
   xs = List @@ pd /. x_X :→ If[PositiveQ[x], Xp[x⟦4⟧, x⟦1⟧], Xm[x⟦2⟧, x⟦1⟧]];
   rots = Table[0, {2 n}];
   front = {0};
   For[k = 0, k < 2 n, ++k,
    If[k == 0 ∨ FreeQ[front, -k],
     front = Flatten[front /. k → Catch[xs /. {
             Xp[k + 1, l_] | Xm[l_, k + 1] :→ Throw[{l, k + 1, 1 - l}],
             Xp[l_, k + 1] | Xm[k + 1, l_] :→ (++rots⟦l⟧; Throw[{1 - l, k + 1, l}])
            }]],
     If[MatchQ[front, {___, k, ___, -k, ___}], --rots⟦k + 1⟧]
    ]
   ];
   RVK[xs, rots]
  ];
RVK[K_] := RVK[PD[K]];
```

In[∘]:= `RVK[Knot[3, 1]]`

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]= RVK[{Xm[4, 1], Xm[6, 3], Xm[2, 5]}, {0, 0, 0, -1, 0, 0}]

In[∘]:=
```
rot[_, 0] = 𝔼[0, 0, 1];
rot[i_, n_Integer] /; n > 0 :=
  rot[i, n] = Module[{j}, (rot[i, n - 1] kCC_j) ~B_{i,j}~ km_{i,j→i}];
rot[i_, n_Integer] /; n < 0 := rot[i, n] = Module[{j}, (rot[i, n + 1] \overline{kCC}_j) ~B_{i,j}~ km_{i,j→i}];
```

In[∘]:= `rot[i, -3]`

Out[∘]= $\mathbb{E}\left[0, 0, \frac{1}{T^{3/2}} + \frac{3\, a_i\, \epsilon}{T^{3/2}} + \frac{9\, a_i^2\, \epsilon^2}{2\, T^{3/2}} + O[\epsilon]^3\right]$

In[∘]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := Z[rvk] = Module[{todo, n, rots, ζ, done, st, x, ζ1, i, j, k, k1, k2, k3},
   {todo, rots} = List @@ rvk;
   AppendTo[rots, 0];
   n = Length[todo];
   ζ = 𝔼[0, 0, 1];
   done = {0};
   st = Range[0, 2 n + 1];
   While[todo =!= {},
    {x} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
    Z$todo = todo; Z$x = x;
    {i, j} = List @@ x;
    ζ1 = Switch[Head[x],
      Xp, m_{j,k→j}[R⁺_{i,j} (R⁻_{k3,k} nr_{k1} ul_{k2} // m_{k,k1→k} // m_{k,k2→k} // m_{k,k3→k})],
      Xm, m_{j,k→j}[R⁻_{i,j} (R⁺_{k,k3} nr_{k1} ul_{k2} // m_{k,k1→k} // m_{k,k2→k} // m_{k,k3→k})]
     ];
    ζ1 = rot[k, rots[[i]]] ζ1 // m_{k,i→i}; rots[[i]] = 0;
    ζ1 = ζ1 rot[k, rots[[i + 1]]] // m_{i,k→i}; rots[[i + 1]] = 0;
    ζ1 = rot[k, rots[[j]]] ζ1 // m_{k,j→j}; rots[[j]] = 0;
    ζ1 = ζ1 rot[k, rots[[j + 1]]] // m_{j,k→j}; rots[[j + 1]] = 0;
    ζ *= ζ1;
    If[MemberQ[done, i], ζ = ζ // m_{i,i+1→i}; st = st /. st[[i + 2]] → st[[i + 1]]];
    If[MemberQ[done, i - 1], ζ = ζ // m_{st[[i]],i→st[[i]]}; st = st /. st[[i + 1]] → st[[i]]];
    If[MemberQ[done, j], ζ = ζ // m_{j,j+1→j}; st = st /. st[[j + 2]] → st[[j + 1]]];
    If[MemberQ[done, j - 1], ζ = ζ // m_{st[[j]],j→st[[j]]}; st = st /. st[[j + 1]] → st[[j]]];
    done = done ⋃ {i - 1, i, j - 1, j};
    todo = DeleteCases[todo, x]
   ];
   ζ /. {u_0 → u, c_0 → c, w_0 → w}
  ]
```

In[∘]:=
```
EndProfile[];
```

Profile

In[∘]:=
```
PrintProfile[]
```

Profile

Out[∘]=
```
ProfileRoot is root. Profiled time: 136.595
  (  136)   0.767/ 107.910 above Bind
  (  126)   0.046/   0.046 above CF
  (   12)       0/   3.094 above Boot[1]
  (   16)   0.031/   7.265 above Boot[2]
  (    5)   0.110/  18.281 above Boot[3]
CF: called 29902 times, time in 73.923/81.155
  ( 28576)    7.232/   7.232 under CF
  (  600)  52.661/  59.893 under LZip
  (  126)   0.046/   0.046 under ProfileRoot
  (  600)  13.984/  13.984 under QZip
  ( 28576)    7.232/   7.232 above CF
Zip: called 1705 times, time in 32.772/145.32
  (  200)   5.744/  27.107 under LZip
```

```
  (  200)   2.514/ 12.264 under QZip
  ( 1305)  24.514/ 105.950 under Zip
  ( 1705)   6.599/  6.599 above Collect
  ( 1305)  24.514/ 105.950 above Zip
LZip: called 200 times, time in 17.876/104.876
  (  200)  17.876/ 104.880 under Bind
  (  600)  52.661/ 59.893 above CF
  (  200)   5.744/ 27.107 above Zip
Collect: called 1705 times, time in 6.599/6.599
  ( 1705)   6.599/  6.599 under Zip
QZip: called 200 times, time in 4.095/30.343
  (  200)   4.095/ 30.343 under Bind
  (  600)  13.984/ 13.984 above CF
  (  200)   2.514/ 12.264 above Zip
Bind: called 200 times, time in 1.047/136.266
  (  136)   0.767/ 107.910 under ProfileRoot
  (   24)   0.078/  3.062 under Boot[1]
  (   24)   0.124/  7.218 under Boot[2]
  (   16)   0.078/ 18.077 under Boot[3]
  (  200)  17.876/ 104.880 above LZip
  (  200)   4.095/ 30.343 above QZip
Boot[3]: called 11 times, time in 0.204/30.561
  (    5)   0.110/ 18.281 under ProfileRoot
  (    6)   0.094/ 12.280 under Boot[3]
  (   16)   0.078/ 18.077 above Bind
  (    6)   0.094/ 12.280 above Boot[3]
Boot[2]: called 18 times, time in 0.047/7.297
  (   16)   0.031/  7.265 under ProfileRoot
  (    2)   0.016/  0.032 under Boot[2]
  (   24)   0.124/  7.218 above Bind
  (    2)   0.016/  0.032 above Boot[2]
Boot[1]: called 20 times, time in 0.032/3.986
  (   12)       0/  3.094 under ProfileRoot
  (    8)   0.032/  0.892 under Boot[1]
  (   24)   0.078/  3.062 above Bind
  (    2)       0/      0 above Boot[0]
  (    8)   0.032/  0.892 above Boot[1]
Boot[0]: called 2 times, time in 0./0.
  (    2)       0/      0 under Boot[1]
```